epreuve(p):={      // simule un événement de probabilité p

si alea(0,1)<p

alors retourne 1

sinon retourne 0

fsi

}:;

echantillon(p,n):={     // simule n épreuves, et compte le nombre de succès obtenus

local k,S;        // un essai est plus rigoureusement appelé "schéma de Bernouilli"

S:=0;

pour k de 1 jusque n 

faire

    S:=S+(epreuve(p))

fpour;

retourne S

};

loi_bin_simulee(p,n,N):={ //  on répéte N essais et l'on effectue une statisitique sur les nombres de succès

local k,L,s;              //  autrement dit, on compte le nombre d'essais où l'on a eu 0 succès, puis un succès etc... on range ces résultats dans une liste L

L:=[0$(n+1)];    // on initialise une liste de n+1 zéros                   

pour k de 1 jusque N 

faire

    s:=echantillon(p,n);     // s est le nombre de succès obtenus lors de l'essai

    L[s]:=L[s]+1;      // on met à jour la liste des résultats

fpour;

L:=L/N;     //on passe des effectifs aux fréquences

retourne L

};

frequences_cumulees(L):={   // cumulation croissante des fréquences

local k,FC,m;

m:=length(L);

FC:=[0$(m)];

FC[0]:=L[0];

pour  k de 1 jusque m-1 faire

  FC[k]:=FC[k-1]+L[k];

fpour;

retourne FC

};

intervalle_fluctu(FC):={  //FC est une liste de fréquences cumulées dont le dernier élément est 1

local k,a,b;              // cet fonction fournit l'intervalle de fluctuation symétrique au seuil de 0.95, lié à 

k:=0;                     // la réalisation de cette expérience aléatoire

tantque FC[k]<=0.025 faire

  k:=k+1

ftantque;

a:=k;

tantque FC[k]<0.975 faire

  k:=k+1

ftantque;

b:=k;

return [a,b]

};

diagramme(L,a,b):={        //Construit le diagramme en bâtons correspondant

local k, n, A, B;

DispG; ClrGraph;

n:=length(L);

couleur(1)

pour k de 0 jusque n-1 

faire

    si k<a ou k>b alors couleur(1) sinon couleur(4);fsi

    A:= point(k,0); 

    B:= point(k,L[k]);

    segment(A,B);

fpour;

};

srand;            //initialisation du calcul aléatoire

n:=20;            // les valeurs de p, n et N peuvent aussi être saisies

p:=1/2;

N:=10000;

LS:=loi_bin_simulee(p,n,N);  // loi simulée

LC:=frequences_cumulees(LS);  

afficher (approx(LS));

afficher (approx(LC));

[a,b]:=intervalle_fluctu(LC);

afficher("intervalle de fluctuation à 0.95  : "+l);

//afficher("intervalle de fluctuation au seuil de 0.95, d'après cette simulation ",IF);

diagramme(LS,a,b);

